


5. Πλεγματικές θέσεις

Θεωρούμε τα παρακάτω αντικείμενα, τα οποία είναι τοποθετημένα σε διάφορα σημεία ενός δισδιάστατου πλέγματος.



Υποθέτοντας ότι το πλέγμα είναι θεωρητικά απείρων διαστάσεων, θέλουμε να εκφράσουμε τη θέση που βρίσκεται καθένα από τα αντικείμενα όχι με χρήση καρτεσιανών συντεταγμένων, αλλά χρησιμοποιώντας σχέσεις που να δηλώνουν που βρίσκεται κάποιο αντικείμενο σε σχέση με τα υπόλοιπα. Έτσι δεν απαιτείται η χρήση αριθμών για την περιγραφή αυτή.

Για το λόγο αυτό θα εισάγουμε τις έννοιες αριστερά (left), δεξιά (right), πάνω (above) και κάτω (below) με προφανή χρήση. Έτσι το κατηγορημα $\text{left}(a,b)$ δηλώνει ότι το αντικείμενο a βρίσκεται στην αμέσως επόμενη προς τα αριστερά πλεγματική θέση σε σχέση με τη θέση που βρίσκεται το αντικείμενο b . Παρόμοια

μπορούμε να εργασθούμε και για τις υπόλοιπες έννοιες (δεξιά, πάνω, κάτω), για όλα τα υπόλοιπα σημεία του πλέγματος. Με βάση τις θέσεις των αντικειμένων στο δοσμένο πλέγμα, έχουμε τα παρακάτω γεγονότα:

left(a,b).

left(d,e).

left(e,f).

left(f,g).

above(a,d).

above(b,e).

above(e,h).

above(h,i).

above(c,g).

Για να ορίσουμε την έννοια δεξιά (right) αρκεί να χρησιμοποιήσουμε τα γεγονότα left και την πληροφορία ότι αντιστροφής δεξιού – αριστερού, δηλαδή αρκεί να δηλώσουμε ότι το αντικείμενο X βρίσκεται στα δεξιά του αντικειμένου Y, αν το αντικείμενο Y βρίσκεται στα αριστερά του X. Άρα:

right(X,Y) :- left(Y,X).

Παρόμοια, μπορούμε να δηλώσουμε την έννοια κάτω (below), αρκεί να δηλώσουμε ότι το αντικείμενο X βρίσκεται κάτω από το αντικείμενο Y, αν το αντικείμενο Y βρίσκεται πάνω από το αντικείμενο X. Έτσι:

below(X,Y) :- above(Y,X).

Στη συνέχεια μπορούμε να δώσουμε ορισμό για το πότε ένα αντικείμενο X βρίσκεται ανάμεσα (between) σε δυο αντικείμενα Y και Z, είτε στην οριζόντια, είτε στην κάθετη διεύθυνση. Για να βρίσκεται το X ανάμεσα στο Y και Z κατά την οριζόντια διεύθυνση, αρκεί να έχει στα δεξιά του το Y και στα αριστερά του το Z.

Παρόμοια, για να βρίσκεται το X ανάμεσα στα Y και Z κατά την κατακόρυφη διεύθυνση, αρκεί να βρίσκεται πάνω από το Y και κάτω από το Z . Έτσι:

`between(X, Y, Z) :- right(X,Y), left(X,Z).`

`between(X, Y, Z) :- above(X,Y), below(X,Z).`

Οι δυο παραπάνω εκφράσεις μπορούν να συγχωνευθούν σε μια με κατάλληλη χρήση της διάζευξης, με χρήση του συμβόλου ; της Prolog.

`between(X, Y, Z) :- (right(X,Y), left(X,Z)); (above(X,Y), below(X,Z)).`

Στο σώμα του παραπάνω κανόνα έγινε χρήση του συμβόλου (;) το οποίο στην Prolog δηλώνει τη διάζευξη προτάσεων. Μπορούμε στη συνέχεια να ορίσουμε μια συνθήκη που να εξετάζει αν κάποιο αντικείμενο X βρίσκεται τοποθετημένο διαγώνια (diagonal) με κάποιο αντικείμενο Y . Για να συμβαίνει αυτό αρκεί το X να βρίσκεται στα δεξιά ή στα αριστερά ενός αντικειμένου Z και με τη σειρά του το αντικείμενο Z να βρίσκεται πάνω ή κάτω από το αντικείμενο Y . Έτσι:

`diagonal(X,Y) :- (right(X,Z); left(X,Z)), (above(Z,Y); below(Z,Y)).`

Τέλος, μπορούμε να ορίσουμε συνθήκη για το αν ένα αντικείμενο X βρίσκεται γενικά στα αριστερά (`on_left`) ενός αντικειμένου Y (δηλαδή σε οποιαδήποτε πλεγματική θέση που βρίσκεται στην ίδια οριζόντια γραμμή με το Y και προς τα αριστερά του Y). Επειδή δεν είναι δυνατόν να γνωρίζουμε στη γενική περίπτωση πόσες είναι οι πλεγματικές θέσεις που απέχει το X από το Y , πρέπει να ορίσουμε αυτή τη συνθήκη με χρήση αναδρομής. Δηλαδή αν η πλεγματική απόσταση των X και Y είναι μεγαλύτερη από 1, τότε να γίνεται κλήση του ίδιου του ορισμού. Φυσικά πρέπει να περιλάβουμε και την περίπτωση που το X βρίσκεται αμέσως αριστερά του Y . Έτσι:

`on_left(X,Y):- left(X,Y).`

`on_left(X,Y):- left(X,Z), on_left(Z,Y).`

Με τον παραπάνω ορισμό έχουμε διασφαλίσει ότι θα πάρουμε ως απάντηση το αντικείμενο X που βρίσκεται αμέσως αριστερά του Y ($1^{ος}$ κανόνας). Επιπλέον, αν συμβεί το X να είναι στα αριστερά ενός σημείου Z και το Z με τη σειρά του είναι στα αριστερά του Y , ($2^{ος}$ κανόνας), τότε η αναδρομική κλήση της `on_left(Z,Y)` εξασφαλίζει ότι η αναζήτηση θα συνεχισθεί για όλες θέσεις προς τα αριστερά του Y , μέχρι να φτάσουμε βήμα-βήμα στο X .

Με τον ίδιο τρόπο μπορούμε να ορίσουμε συνθήκες για τον εντοπισμό ενός αντικειμένου X που βρίσκεται γενικά κάπου στα δεξιά (`on_right`) ενός αντικειμένου Y , δηλαδή βρίσκεται στην ίδια οριζόντια ευθεία με το Y , αλλά στα δεξιά του Y . Είναι:

`on_right(X,Y):- right(X,Y).`

`on_right(X,Y):- right(X,Z), on_right(Z,Y).`

Αντίστοιχα μπορούμε να ορίσουμε συνθήκες για τον εντοπισμό αντικειμένου X που βρίσκεται σε οποιαδήποτε θέση πάνω (`upper`) από ένα αντικείμενο Y (δηλαδή το X βρίσκεται στην ίδια κάθετη ευθεία με το Y και σε θέση πάνω από αυτή του Y). Έτσι:

`upper(X,Y):- above(X,Y).`

`upper(X,Y):- above(X,Z), upper(Z,Y).`

Τέλος μπορούμε να ορίσουμε συνθήκες για τον εντοπισμό αντικειμένου X που βρίσκεται σε οποιαδήποτε θέση κάτω (`lower`) από ένα αντικείμενο Y (δηλαδή το X βρίσκεται στην ίδια κάθετη ευθεία με το Y και σε θέση κάτω από αυτή του Y). Έτσι:

`lower(X,Y):- below(X,Y).`

`lower(X,Y):- below(X,Z), lower(Z,Y).`